
 Abstract 
Stereo Imaging is a powerful yet seldom utilized technique for determining the distance to 
objects using a pair of camera spaced apart.  This is fundamentally the same visual system 
used by humans and most other animals.  The extremely high computational requirements 
of stereo vision limit application to non-realtime applications or to applications where high 
computational horsepower is available.   

In a typical application a pair of cameras is spaced apart at roughly the same spacing as 
human eyes.  Each camera images the scene from a slightly different angle, and thus objects 
are imaged onto a slightly different region of the image sensor.  For objects close to the 
camera the difference in angle to each sensor is larger than for objects far from the camera 
and thus there is a larger disparity in the image. 

This sample illustrates a simple yet effective method of determining the disparity between 
pixels in two images using a block matching technique to determine correlation between 
groups of pixels in a stereo pair.   

Theory 
The basic structure of a stereo vision application is shown in figure 1.   Two cameras are 
spaced apart by the baseline distance B.  Each camera images the object, but from a slightly 
different angle.  The distance to the object can then be computed by: 

d
BfD =  

Where D is the distance to the object, B is the baseline distance between the stereo imagers, f  
is the focal length of the camera, and d is the disparity.  d is the difference in location of the 
image of the object between the Left and Right images.  Finding d by determining the 
correspondence between pixels in the stereo pair is the primary problem to be solved by this 
application.  Our algorithm will compute d in units of pixels, which can be converted to 
distance units by multiplying by the pixel size.  Also, this simple formula assumes idealized 
optics.   

Chart 1 shows the relationship between pixel disparity and the distance to an object.  Note 
the logarithmic distance scale.  This chart was made using the parameters for a Point Grey 
Research Bumblebee2 stereo camera with a baseline distance (B) of 120mm, a focal length (f) 
of 3.8mm and a pixel pitch of 4.65µm. 

A simple block-matching algorithm is used to find the disparity (d). Figure 2 shows an image 
of a flower in both images in the pair.  The left image serves as the reference image.  A block 
of pixels from the left image (denoted in yellow) is translated horizontally across the right 
image.  The goal is to find the best match between the block of pixels in the left image a 
block in the right image.  The number of pixels the block must be translated to find the best 
match is the disparity value. 

There are several different methods of determining the ‘best’ match.  In this example 
application the sum-of-squared differences (SSD) is used.  The position with the lowest SSD 
is the disparity.  Sum of absolute differences (SAD) is often commonly used because of the 
lower computational complexity.  However, with GPUs the SSD can be computed in the 
same time as the SAD so SSD was chosen. 

The size of the block is also a critical parameter for the performance of the algorithm.  
Larger blocks can provide better correspondence results but require more computation.  



Stereo Imaging with CUDA 

    

 

 

January 2008  Page 4 of 20 

 
As one can quickly see, the computation requirements required for stereo disparity 
computation can be enormous.  The above computation may be done on each pixel in the 
reference image.  A typical disparity range may be 50 – 100 pixels (sometimes done in sub-
pixel steps).  For a VGA (640x480) image, with an 11x11 block size and a 50 pixel disparity 
range there are almost 2 billion differences taken for a single image pair, then those 
differences are squared and accumulated!   

It should be noted that the above discussion assumes the imagers are aligned and free from 
optical distortion.  This is rarely the case.  A variety of known methods may be used during 
manufacture and operation to calibrate the mechanical alignment of the camera and remove 
optical distortions. 

Of course, in order to effectively implement real-time stereo correlation – even on an 
NVIDIA GPU - some clever optimizations must be made to achieve real time (30 fps) 
performance.   

Note:  
As seen by the graph the distance changes dramatically for small disparities and only 
slightly with large disparities.  Thus to detect distant objects high resolution is needed and 
sub-pixel disparity steps are advantageous.  For closer objects, however, a lower resolution 
can be used, effectively increasing the pixel pitch and reducing the number of block 
matches which must be performed.  It is not suitable to have disparity steps greater than 1 
pixel without first reducing the resolution of the image, otherwise aliasing may result. 
 
To achieve higher performance over a large disparity range it is common to use image 
pyramids.  An image pyramid (also called MIPMAPS in the computer graphics world) is a 
group of successively lower resolution images, typically at a 2-to-1 scale.  The lower 
resolution images can be processed for nearby objects and the higher resolution image can 
be processed for more distant objects.  It is also common to use a selected region of 
interest in each image.   

This sample implementation can readily be adopted to use image pyramids by creating 
multiple images and calling the stereo kernel on the different images with different disparity 
ranges and regions-of-interest. 

Strategy 
Even for the simple block-matching algorithm described above there are many different 
approaches to implementation on the GPU.  The strategy employed in this example is highly 
optimized, but certainly not guaranteed to be the only fast approach.  This white paper will 
only describe the method used this example rather than to explore all of the different 
approaches or contrasting the advantages of each method.    

The primary goal of this example is to be very fast, with a secondary goal of being fairly 
flexible to allow changing of parameters.    

Here are some critical guidelines for optimizing this application: 

 Avoid obscenely redundant computation – Many computations preformed for 
one pixel can also be used by neighbors 

 Keep global memory Coalesced 
 Minimize global memory reads/writes 
 Exploit texture hardware (especially for sub-pixel disparity computations) 
 Create enough threads & thread blocks to keep the processors busy 

 
 The first strategic decision is where to store the image.  Texturing from a CUDA Array is 
used for both the reference and comparison image.  Using texturing provides several 
advantages: 

 No coalescence requirements and fast, cached access 
 Boundary clamping 
 Bilinear interpolation (for sub-pixel disparity measurements) 
 Changing image source data type is trivial and does not require re-optimization 

of the algorithm  
 

The next step is to determine how to allocate threads to the problem.  The approach chosen 
uses a thread to process a column of pixels.  This is a departure from the common approach 
of using a thread per pixel but, as will be seen, this allows us to eliminate some redundant 
computation and minimize the impact of kernel size on overall performance.   

Figure 3 illustrates the overall scheme using a 5x5 kernel comparison size and a 16x1 thread 
block size (in the application a thread block size of at least 32 is used, with 64-128 being 
optimal).   Each thread sums the squared differences of a column of pixels the height of the 
kernel.  It accumulates squared differences between the pixels in the reference image (left) 
and the comparison image (right); the comparison image being offset by the disparity value.  
The values of the pixels in each image are retrieved using texture instructions.  The sum of 
squared differences for each column is stored in local shared memory.  Because two extra 
pixels are used on each side of the processed block for the kernel radius, four threads must 
make an extra read and summation (threads 0 – 3 in the figure). 

After the column sum-of-squared differences are completed for the block of threads then 
each thread sums the column SSD values from the neighboring columns within the kernel 
radius to determine the total SSD for the entire kernel.  This value is tested to determine if 
the current disparity value is the best correspondence match for the present pixel.  Three 
pixels have been highlighted for illustration in green, blue & orange with the associated 
kernel pixels outlined in a dashed line of the same color. 

Mathematically the SSD computation is expressed as: 

Where x & y are the current pixel coordinates, RH is the horizontal kernel radius, RV is the 
vertical horizontal radius, and d is the current disparity offset being evaluated.  The kernel 
radius is the number of pixels to the left, right, top, or bottom of the pixel in the block 
matching kernel (RH = 2 & RV = 2 for a 5x5 kernel). 

After the first row of pixels has been processed by a thread block subsequent rows can be 
processed with increased efficiency (particularly for large kernels).  A rolling window scheme 
is used.  Rather than repeat the summation of all the squared differences in the column, the 
squared difference of the pixels in the first row is subtracted from the previous column sum, 
and then the squared difference of pixels in a new row is added to the column num.  This 
value is equivalent to re-summing the rows involved, but requires only two squared 
difference computations and two pixel reads from each image.  This rolling window 
computation continues until SSD is determined for all the rows allocated to a thread. 

The entire process described above is repeated for each disparity step (d).  The use of texture 
reads to retrieve pixel values allows use of sub-pixel disparity steps for improved accuracy.  
At each disparity step, the SSD value determined for the kernel is compared with the current 
minimum SSD value computed from prior disparity steps.  The current minimum SSD result 
and the corresponding d value are stored in global memory in arrays the size of the image.  If 
the newly computed SSD value is less than the previous minimum SSD value, this new value 
becomes the minimum and is stored along with the corresponding d value in memory. 

At the end of this process a disparity value indicative of the best correspondence has been 
computed and stored in global memory. 

Note: Shared Memory 
This example highlights the power of shared memory and data parallel programming.  
The column SSD values are stored in shared memory which is instantaneously 
available to all threads within a block.  This allows a significant amount of data 
sharing between threads and thus a dramatic improvement in efficiency.  Rather than 
each thread computing all the individual squared-difference values within a kernel the 
threads cooperate to compute pieces of the SSDs for each other.   Each thread 
computes a column sum and stores it in shared memory. Neighboring threads then 
use that column sum for their computation.  Furthermore, these column sums are 
preserved in shared memory for the rolling-window summations for the remaining 
rows. 
Using a traditional brute-force approach each 5x5 kernel would require 25 squared 
difference computations and 24 additions.  In this example, after the initial 5 rows 
there are only 2 squared-difference computations and 4 additions (plus a few extra 
for the extra apron pixels).  For a more typical 11x11 kernel, there would be 2 
squared differences + 10 additions, rather than 121 squared-differences and 120 
additions – a 20X reduction in the number of operations! 

Code Walkthrough 
In this section we will walk line-by-line through the critical code in this example.  We will 
not examine the supporting code in main.cpp which contains image acquisition and drawing 
code.  Rather we sill start with the function stereoProcess in stereo.cu: 

First there are a number of global variables used to store pointers to data and a few 
parameters.   

The Image width & height. 
static int g_w; 
static int g_h; 

Pointers to memory for the disparity value (d) and the current minimum SSD, and the data 
pitch; also on the GPU:  
static float *g_disparityLeft; 
static int *g_minSSD; 
static size_t g_floatDispPitch; 

 

Pointers to cudaArrays, which contain a copy of the data for texturing 
cudaArray * g_leftTex_array; 
cudaArray * g_rightTex_array; 

 

These values store OpenGL buffer ID’s which are used to draw the image on the screen 
using CUDA’s OpenGL interop capability.  These are set from the main program after the 
bufferes are created.. 
unsigned int LeftImage_GLBufferID; 
unsigned int RightImage_GLBufferID; 
unsigned int DisparityImage_GLBufferID; 

 

Declaration of the textures used for accessing the image data.  The parameter 
cudaReadModeNormalizedFloat indicates that a texture lookup instruction will 
return a floating point value from 0 to 1.0 rather than an 8-bit unsigned char.  This allows 
for easy changing of the source data type if necessary.   
texture<unsigned char, 2, cudaReadModeNormalizedFloat> leftTex; 
texture<unsigned char, 2, cudaReadModeNormalizedFloat> rightTex; 

 

There are also a couple of defined constants: 
#define ROWSperTHREAD 40   // the number of rows a thread will process 
#define BLOCK_W 64    // the thread block width 
#define RADIUS_H 5    // Kernel Radius 5V & 5H = 11x11 kernel  
#define RADIUS_V 5 
#define MIN_SSD 500000       // The mimium acceptable SSD value 
#define STEREO_MIND 0.0f   // The minimum d range to check 
#define STEREO_MAXD 50.0f  // the maximum d range to check  
#define STEREO_DISP_STEP 1.0f // the d step, must be <= 1 to avoid aliasing 
#define SHARED_MEM_SIZE ((BLOCK_W + 2*RADIUS_H)*sizeof(int) ) // amount of  
                                                             //shared memory used 

The function SetupStereo allocates GPU memory and sets critical parameters in stored 
in global variable. 

We pass the width & height into the function and store these globally.    
extern "C" void SetupStereo(unsigned int w, unsigned int h) 

{ 
g_w = w; 
g_h = h; 

 

Next, memory must be allocated on the GPU for the disparity value and the mimimum SSD 
value.  Using cudaMallocPitch insures properly aligned memory and 
g_floatDispPitch indicates the number of bytes between sequential rows in an image.  
Since we will address these variables as float’s we divide the number of bytes by 
sizeof(float). 

cudaMallocPitch((void**)&g_disparityLeft,&g_floatDispPitch,w*sizeof(float),h));   
cudaMallocPitch((void**)&g_minSSD,g_floatDispPitch,w*sizeof(int),h)); 
 
g_floatDispPitch /= sizeof(float); 

 

Next, allocate cudaArrays for the images so we can access using textures.  If the data type 
is not unsigned char, this must be changed accordingly. 

cudaChannelFormatDesc U8Tex = cudaCreateChannelDesc<unsigned char>();  
cudaMallocArray(&g_leftTex_array, &U8Tex, g_w, g_h);  
cudaMallocArray(&g_rightTex_array, &U8Tex, g_w, g_h); 

 

Finally, we print out the total and free memory on the GPU.  This is useful for debugging.  
Also, this is one case where is it acceptable to mix a driver API function (cuMemGetInfo)  
within the context of a runtime API program. 

int free, total; 
cuMemGetInfo(&free,&total); 
printf("Memory After Allocation - Free: %d, Total: %d\n", 

     free/(1024*1024),total/(1024*1024)); 

} 

Note: It is also an option to use pinned host memory using cudaMallocHost. Pinned 
memory is not managed by the operating systems virtual memory system and 
remains fixed in physical memory.  Transfers from the CPU to the GPU memory are 
about twice as fast when pinned memory is used.  In this example VFW and Point 
Grey APIs are used as video sources, neither of which can conveniently support the 
use of pinned memory allocated by the CUDA driver.  However, the user is 
encouraged to try using pinned memory if other video sources are available. 

 

The main function called for every frame is stereoProcess.  Pointers to the left and 
right images are provided as parameters. 
extern "C" float stereoProcess(unsigned char * p_hostLeft, unsigned char * p_hostRight) 
{ 

unsigned char * GLtemp; 

 

Set up a timer for evaluation: 

unsigned int timer; 
CUT_SAFE_CALL(cutCreateTimer(&timer)); 
CUT_SAFE_CALL(cutStartTimer(timer));   

 

Dimensions of the thread blocks and grid. 
dim3 grid(1,1,1); 
dim3 threads(16,8,1); 

 

The next step is to copy the image from the host onto the GPU.  If the image is coming 
from an AVI file in Windows it must be first converted from a BGR color representation to 
a Grayscale representation and also inverted vertically.  In this block of code, the color image 
is copied to memory and converted to a grayscale image in a kernel function.  The output of 
this kernel is then copied into a cudaArray for texturing.  If this image is grayscale, it is 
simply copied directly form the host into a cuda array. 

if(ImageType == IMAGE_TYPE_AVI_DIB) { 
     uchar4 * tempColorImage; 
     unsigned char * tempGrayImage; 
     size_t color_pitch; 
     size_t gray_pitch; 
     grid.x = divUp(g_w,threads.x*4); 
     grid.y = divUp(g_h,threads.y);   
      
     cudaMallocPitch((void**)&tempColorImage,&color_pitch,g_w*sizeof(uchar4),g_h); 
     cudaMallocPitch((void**)&tempGrayImage,&gray_pitch,g_w,g_h); 
      
     // Left 
     cudaMemcpy2D(tempColorImage,color_pitch,p_hostLeft,g_w*sizeof(uchar4), 
                  g_w*sizeof(uchar4),g_h,cudaMemcpyHostToDevice); 
     convertBGRA_Invert_toGSKernel<<<grid,threads>>>(tempColorImage,tempGrayImage, 
                                  color_pitch/sizeof(uchar4),gray_pitch,g_w,g_h); 
   cudaThreadSynchronize(); 
     cudaMemcpy2DToArray(g_leftTex_array,0,0,tempGrayImage, 
                         gray_pitch,g_w,g_h,cudaMemcpyDeviceToDevice); 
   
  // Right 
     cudaMemcpy2D(tempColorImage,color_pitch,p_hostRight,g_w*sizeof(uchar4), 
                  g_w*sizeof(uchar4),g_h,cudaMemcpyHostToDevice); 
   convertBGRA_Invert_toGSKernel<<<grid,threads>>>(tempColorImage,tempGrayImage, 
                                  color_pitch/sizeof(uchar4),gray_pitch,g_w,g_h); 
   cudaThreadSynchronize() 
     cudaMemcpy2DToArray(g_rightTex_array,0,0,tempGrayImage, 
                         gray_pitch,g_w,g_h,cudaMemcpyDeviceToDevice); 
   
   cudaFree(tempColorImage); 
   cudaFree(tempGrayImage); 
}  
else {  
// Greyscale Image, just copy it. 
    cudaMemcpyToArray(g_leftTex_array, 0, 0, p_hostLeft,g_w * g_h,       
                       cudaMemcpyHostToDevice); 
   cudaMemcpyToArray(g_rightTex_array, 0, 0, p_hostRight,g_w * g_h,               
                        cudaMemcpyHostToDevice); 
}      

 

To display the image a copy of the grayscale image is copied into a OpenGL pixel buffer 
object. 

cudaGLMapBufferObject( (void**)&GLtemp, LeftImage_GLBufferID);  
cudaMemcpyFromArray(GLtemp,g_leftTex_array,0,0,g_w*g_h,cudaMemcpyDeviceToDevice); 
cudaGLUnmapBufferObject(LeftImage_GLBufferID); 
cudaGLMapBufferObject( (void**)&GLtemp, RightImage_GLBufferID);  

cudaMemcpyFromArray(GLtemp,g_rightTex_array,0,0,g_w*g_h,cudaMemcpyDeviceToDevice); 
cudaGLUnmapBufferObject(RightImage_GLBufferID); 

 

Texture parameters are set so bilinear interpolation will be preformed for inter-pixel 
lookups.  Then the cudaArrays containing the images are bound to the textures. 

// Set up the texture parameters for bilinear interpolation & clamping 
leftTex.filterMode = cudaFilterModeLinear; 
cudaBindTextureToArray(leftTex, g_leftTex_array); 
rightTex.filterMode = cudaFilterModeLinear; 
cudaBindTextureToArray(rightTex, g_rightTex_array); 

 

Here’s where we kick off the kernel.  As discussed above we use a 1-dimentional block of 
threads to process the several columns of the image.  BLOCK_W is a constant which 
specifies how many threads will cooperate in processing a portion of the image.  This should 
be at least 32 (the warp size) and should be a multple of 32.  There are several factors which 
influence the ideal block width.  The larger the block width, the smaller fraction of the total 
computation will be allocated to the border pixels requiring an extra thread execution.  If the 
block width is small, and the kernel is large, this may have a significant impact.  However, it’s 
important to have many thread blocks available to run on the GPU to keep the SMs busy 
during memory accesses and thread synchronizes, so too large of a block width may be 
detrimental.  Additionally, too large of a block width may cause more frequenty cache misses 
by the texture hardware since a larger portion of the overall image in addressed by a thread 
block.  Block width’s of between 64 and 128 threads are optimal on 8800 series hardware. 

The divUp function is defined elsewhere and simply rounds up the division to be sure 
there are enough blocks to cover the image.  ROWSperTHREAD is another constant which 
sets the number of rows a thread will process for the rolling window strategy.  As with the 
BLOCK_W, having more rows reduces the computation from boundry pixels but too many 
rows may result in an insufficient number of thread blocks or cache ineffeciency. 

SHARED_MEM_SIZE is set the amout of shared memory in bytes per thread block.  It is 
equal to the block width plus two times the kernel radius times four (sizeof(float)). 

threads.x = BLOCK_W; 
threads.y = 1; 
grid.x = divUp(g_w, BLOCK_W); 
grid.y = divUp(g_h,ROWSperTHREAD); 
 
stereoKernel<<<grid,threads,SHARED_MEM_SIZE>>>(g_disparityLeft,g_sumSSD, 

                g_w,g_h,g_floatDispPitch);  

 

Wait for the kernel to complete, unbind the textures and stop the timer.  We are only timing 
the memory transfer and computation but not factoring in the drawing. 

cudaThreadSynchronize(); 
 

cudaUnbindTexture(leftTex); 
cudaUnbindTexture(rightTex); 
 
CUT_SAFE_CALL(cutStopTimer(timer));  // don't time the drawing  
float retval = cutGetTimerValue(timer); 

 

The remaining code in this function simply takes the computed disparity map in 
g_disparityLeft and draws it as an RGB false color image.  This image is then copied 
to an OpenGL pixel buffer object 

// Now draw the disparity map 
threads.x = 16; 
threads.y = 8; 
grid.x = divUp(g_w , threads.x); 
grid.y = divUp(g_h , threads.y); 
uchar4 * DisparityMap; 
size_t DispPitch; 
cudaMallocPitch((void**)&DisparityMap,&DispPitch,g_w*sizeof(uchar4),g_h);  
  
drawDisparityKernel<<<grid,threads>>>(DisparityMap,DispPitch/sizeof(uchar4), 
                          g_disparityLeft,g_floatDispPitch,g_w,g_h);  
cudaThreadSynchronize(); 
 
// now copy the output for openGL 
cudaGLMapBufferObject( (void**)&GLtemp, DisparityImage_GLBufferID);  
cudaMemcpy2D(GLtemp,g_w*sizeof(uchar4),DisparityMap,DispPitch, 
             g_w*sizeof(uchar4),g_h,cudaMemcpyDeviceToDevice); 
cudaGLUnmapBufferObject(DisparityImage_GLBufferID); 
cudaFree(DisparityMap); 
return retval; 

 

Now for the main stereo kernel:  There are five parameters:  disparityPixel & 
disparityMinSSD point to memory containing the disparity value (d) and the current 
minimum sum-of-squared-difference values for each pixel.  width & height are the 
image width & height, and out_pitch specifies the pitch of the output data in words (i.e. 
the number of floats between the start of one row and the start of the next.). 
__global__ void stereoKernel( float *disparityPixel,  

        float *disparityMinSSD, 
         int width, 
        int height, 
        size_t out_pitch)   
{  

 

The shared memory array col_ssd stores the sum-of-squared-difference values for each 
column.  This shared memory can be accessed by all threads within the block and thus 
allows for cooperation in the computation of the total SSDs as described above.  The 
remainder of the variables are local to a thread.  

extern __shared__ int col_ssd[]; // column squared difference functions 
float d;     // disparity value 
int diff;  // difference temporary value 
int ssd;   // total SSD for a kernel 
float x_tex; // texture coordinates for image lookup 
float y_tex; 
int row;  // the current row in the rolling window  
int i;       // for index variable   

 

The total number of registers used by a kernel program may limit the number of threads an 
SM can run at a given time.  In order to maximize occupancy it is sometimes useful to 
reduce register usage.  The #define statements replace the use of a register variable to store 
X & Y while still keeping the code reasonably clean.   

Note: the number of registers used can be determined by using the –keep nvcc 
compiler option and examining the .cubin file produced during compilation.  The 
CUDA_occupancy_calculator.xls worksheet included with the CUDA SDK will aid in 
determining if register utalization is a concern. 

// use define’s to save registers 

    #define X (__mul24(blockIdx.x,BLOCK_W) + threadIdx.x) 
    #define Y (__mul24(blockIdx.y,ROWSperTHREAD))  

 

Some of the threads need make two reads and squared-difference computations to account 
for the extra pixels needed for the kernel of the border pixels.  Threads with a thread index 
less-than twice the kernel radius must preform double duty.  extra_read_val is set 
greater than zero and indicates the offset into shared memory for these double-duty threads. 

// for threads reading the extra border pixels, this is the offset 
// into shared memory to store the values 
int extra_read_val = 0;   
if(threadIdx.x < (2*RADIUS_H)) extra_read_val = BLOCK_W+threadIdx.x;    

 

Now the global memory used to store the disparity and the mimimum SSD value is 
initialized.  The constant MIN_SSD can be used to set a minimum quality threshold.  If the 
SSD at any value of d is not lower than MIN_SSD, then the the disparityPixel value 
remaines at -1 indicative of no suitable result.  After initiazation threads must be 
synchronized before continuing execution to insure that non-initialized values are not used 
by other threads later during execution. 

// initialize the memory used for the disparity and the disparity difference 
 if(X<width ) 
 { 
  for(i = 0;i < ROWSperTHREAD && Y+i < height;i++) 
  {     
   disparityPixel[__mul24((Y+i),out_pitch)+X] = -1; 
   disparityDiff[__mul24((Y+i),out_pitch)+X] = MIN_DISP;  
  } 

}  
__syncthreads(); 

Note: __syncthreads() is a very important step in data parallel programming when using 
a cooperative thread array.  The GPU thread scheduling mechanism is very complex 
and no assumptions can be safetly made about when the threads within a thread 
block will be completed.  Thus, when one thread within a block depends on a 
computation from another thread within the block synchronization is mandatory.   
 
Synchronization may stall the SM while waiting for some of the threads to complete 
resulting in ineffeciency. For this reason it is especially important to keep a number of 
thread blocks running on and SM at a time when there are frequent synchronization 
points in order to keep the device occupied while one thread block is stalled. 
 
It is also worth mentioning that thread synchronization is one aspect where the CUDA 
emulation mode may differ from running code on a device.  The emulator serializes 
all threads in order to run them on a CPU and thus synchronization is emplicit in 
some instances.  The same code may not run properly on a GPU device if a 
__syncthreads() call was omitted at a critical point. 

Insure that the computation is within the image region and then iterate over the range of 
disparity values to test.  At the start of every disparity interation the col_ssd values are 
initialzed to zero. 

if( X < (width+RADIUS_H) && Y <= (height) )  
{ 

  x_tex = X - RADIUS_H; 
  for(d = STEREO_MIND; d <= STEREO_MAXD; d += STEREO_DISP_STEP) 
  {  
   col_ssd[threadIdx.x] = 0;  

   if(extra_read_val>0) col_ssd[extra_read_val] = 0;     

 

The first step is to accumulate the column sums for the first 2*RADIUS_V+1 rows.  The 
x_tex & y_tex values are set for the thread according to which pixels from the source 
images are to be read.  The x_tex value in the right (comparison) image is offset by the 
current disparity iteration value d.  After each row in this loop the y_tex value is 
incremented.  Once again we must synchronize the threads to ensure all the column SSDs 
are computed before summing up the kernel total. 

In this block the result of the tex2D function is multiplied by 255.0 and converted back into 
an integer.  This is to avoid floating point roundoff error which occurs when adding and 
subtracing values from the column sum accumulator.  In order to enable bilinear 
interpoalation in the texutre unit, the texture read mode must be 
cudaReadModeNormalizedFloat and thus we are required to convert the data back to 
an integral data type. 
   // do the first rows 
   y_tex = Y - RADIUS_V; 
      
   for(i = 0; i <= 2*RADIUS_V; i++) 
   {    

   diff = (int)(255.0f*tex2D(leftTex,x_tex,y_tex)) – 
                      (int)(255.0f*tex2D(rightTex,x_tex-d,y_tex));    
     
    col_ssd[threadIdx.x] += SQ(diff);     
    if(extra_read_val > 0)  
    {         

     diff = (int)(255.0f*tex2D(leftTex,x_tex+BLOCK_W,y_tex)) – 
                    (int)(255.0f*tex2D(rightTex,x_tex+BLOCK_W-d,y_tex)); 

        
     col_ssd[extra_read_val] += SQ(diff);     
    } 
    y_tex += 1.0f; 
   } 
   __syncthreads(); 

 

Now the column sums can be added together in the horizontal direction to compute the 
total SSD for the thread.  This value is compared with the current minimum SSD to 
determine if the current d gives the best correspondence thus far.  After this computation 
threads must by synchronized before the column sums can be updated for the next row. 
   // now accumulate the total     
   if(X < width && Y < height) 
   { 
    ssd = 0; 
    for(i = 0;i<(2*RADIUS_H);i++) 
    { 
     ssd += col_ssd[i+threadIdx.x];     
    } 
    if(ssd < disparityMinSSD[__mul24(Y,out_pitch) + X])     
    { 
     disparityPixel[__mul24(Y,out_pitch) + X] = d; 
     disparityMinSSD[Y*out_pitch + X] = ssd;      
    }    
   }    
   __syncthreads(); 

 

The remainder of the kernal program implements the rolling window scheme to compute 
the SSD for the remainder of rows assigned the the thread block.  First, we must subtract 

the value of the top row in the window from the column sum, then we add in the new 
squared differene value of a new row, which is 2*RADIUS_V + 1 rows from the current 
y_tex value.  The double-duty threads repeat this and y_tex is incremented for the next 
row iteration.  Again, syncthreads before continuing. 
   // now do the remaining rows 
   y_tex = (float)(Y - RADIUS_V); // this is the row we will remove  
   for(row = 1;row < ROWSperTHREAD && (row+Y < (height+RADIUS_V)); row++) 
   {      

   // subtract the value of the first row from column sums     
    diff = (int)(255.0f*tex2D(leftTex,x_tex,y_tex)) –  

                  (int)(255.0f*tex2D(rightTex,x_tex-d,y_tex));   
    col_ssd[threadIdx.x] -= SQ(diff);  
       
    // add in the value from the next row down 

    diff = (int)(255.0f*tex2D(leftTex,x_tex,y_tex +  
                  (float)(2*RADIUS_V)+1.0f)) – 
                  (int)(255.0f*tex2D(rightTex,x_tex-d,y_tex + 
                  (float)(2*RADIUS_V)+1.0f));   

            col_ssd[threadIdx.x] += SQ(diff);    
    
    if(extra_read_val > 0)  

             {           
    diff = (int)(255.0f*tex2D(leftTex,x_tex+(float)BLOCK_W,y_tex)) – 
                      (int)(255.0f*tex2D(rightTex,x_tex-d+(float)BLOCK_W,y_tex));  
    col_ssd[threadIdx.x+BLOCK_W] -= SQ(diff); 
           
      diff = (int)(255.0f*tex2D(leftTex,x_tex+(float)BLOCK_W,y_tex + 
                      (float)(2*RADIUS_V)+1.0f)) – 
                      (int)(255.0f*tex2D(rightTex,x_tex-d+(float)BLOCK_W,y_tex + 
                      (float)(2*RADIUS_V)+1.0f));      
    col_ssd[extra_read_val] += SQ(diff);   
   } 
   y_tex += 1.0f; 
   __syncthreads(); 

 

The following code continues the row loop by adding the column SSDs to produce the total 
SSD for the thread.  As above it is compared to the minimim SSD to see if it is the best 
match thus far.  This row loop continues until the rolling window has processed 
ROWSperTHREAD rows.  This entire procedure repeats in the d for loopuntil the entire 
disparity range is searched.  When complete the global memory disparityPixel will containe 
the value of d with the best correspondence for all the pixels examined. 
    if(X<width && (Y+row) < height) 
    {   
     ssd = 0; 
     for(i = 0;i<(2*RADIUS_H);i++) 
     { 
      ssd += col_ssd[i+threadIdx.x];      
     } 
     if(ssd < disparityMinSSD[__mul24(Y+row,out_pitch) + X])    
  
     { 
      disparityPixel[__mul24(Y+row,out_pitch) + X] = d; 
      disparityMinSSD[__mul24(Y+row,out_pitch) + X] = ssd;    
   
     }    
    } 
    __syncthreads();  // wait for everything to complete 
   } // for row loop 
  } // for d loop 
 } // if 'int the image' loop  

} 

Running the Sample 
The sample code can operate using either a Point Grey Research bumblebee stereo camera 
or by using a pair of video files taken using a stereo camera setup. 

To build the sample for a Point Grey camera the user must supply the appropriate Point 
Grey libraries which come with the Point Grey SDK. NVIDIA does not redistribute Point 
Grey’s software.  The libraries digiclops.lib and triclops.lib from Point Grey must be added 
to the Linker->Input->Additional Dependencies section of the project properties.  
Furthermore the symbol ALLOW_POINT_GREY must be defined in main.cpp 

To run the sample using a Point Grey camera, simply use the command line: 

StereoCamera.exe –c w h 
 

Where w & h are the width & height.   

 

To run the sample using files use the command line: 

StereoCamera.exe –f left_filename right_filename 
 

The sample uses Microsoft’s Video For Windows SDK to open video files and return the 
frames.  The video files must have the appropriate codec’s installed on the machine in order 
to operate.  Also, the left and right video files must be of the same size, and have a frame-by-
frame synchronization for the stereo pair. 

One good source of stereo sample data is Microsoft Research Cambridge: 
http://research.microsoft.com/vision/cambridge/i2i/DSWeb.htm. 

 

Of course the user may also provide stereo image data from another source.  The 
SetupStereo function must be called with the appropriate width & height.  The data 
must be in 8-bit unsigned char format and copied to the memory locations pointed to by 
leftImg & rightImg in main.cpp prior to calling stereoProcess each cycle.   

